

STEP 2: Set up how to get from room to room.

Have you created a map of rooms in your mansion? Great! (If you haven't, go do it now.)

Today, you will create commands so the player can move from room to room in your mansion.

To move around the mansion, you will use these directional commands:

NORTH, SOUTH, EAST, WEST

Make a procedure called `SETUP` in the Logo Editor that will create your mansion and everything in it.

```
TO SETUP
```

```
END
```

The `SETUP` procedure should include an instruction that stores the names of the rooms in a variable, like this:

```
MAKE "ROOMS [ENTRY DINING_ROOM LIBRARY ...]
```

Choose the room that the player starts in. In our mansion, the room is called `ENTRY`.

Put this instruction in the `SETUP` procedure, using the name of your starting room:

```
MAKE "ROOM "ENTRY
```

Your `SETUP` procedure now looks something like this:

```
TO SETUP
  MAKE "ROOMS [ENTRY DINING_ROOM LIBRARY ...]
  MAKE "ROOM "ENTRY
END
```

Create a new procedure named `SETUP_MAP`. It will be called by the `SETUP` procedure.

```
TO SETUP_MAP
```

```
END
```

```
TO SETUP
  SETUP_MAP
  MAKE "ROOMS [ENTRY DINING_ROOM LIBRARY ...]
  MAKE "ROOM "ENTRY
END
```

In the `SETUP_MAP` procedure, you will create a property list for each room. The property list will have a direction word followed by the room you get to if you type that direction. You will use the `PPROPS` command, a handy way to store multiple properties in a property list.

The name of the property list we will create first is **"ENTRY**, the name of the room.

Look at our mansion map. There is only one way to leave the ENTRY. If you go North from the ENTRY, you get to the HALL. If you try to go anywhere else, nothing is there.

```
PPROPS "ENTRY [NORTH HALL EAST NOTHING SOUTH NOTHING WEST NOTHING]
```

The property list has the label NORTH with a value of HALL. All the other directions have a value of NOTHING.

The Living_Room has rooms in all directions, so its property list will look like this:

```
PPROPS "LIVING_ROOM [NORTH GAME_ROOM EAST LIBRARY SOUTH HALL WEST DINING_ROOM]
```

Do you see how that works? If you go North from the Living Room, you get to the Game_Room. If you go East, you get to the Library. If you go South, you go back to the Hall. If you go West, you go to the Dining_Room.

Figure out the property list for each room in your mansion and put them all in the `SETUP_MAP` procedure. You'll have a separate line for the property list for each room with its list of properties showing the direction words that will take you to a different room. Each direction word has either the name of a room or NOTHING. Make sure there is an even number of items in your property list.

The property list for each room will eventually contain many different properties.

Now that you have defined where the player can go from each room, you need to create procedures for those direction commands. The directional commands will use a `GO_ROOM` procedure that you will write in just a moment, after you have created these four procedures. Type these procedures into the editor.

```
TO NORTH
  GO_ROOM "NORTH
END
```

```
TO EAST
  GO_ROOM "EAST
END
```

```
TO SOUTH
  GO_ROOM "SOUTH
END
```

```
TO WEST
  GO_ROOM "WEST
END
```

The `GO_ROOM` procedure takes a direction as input, as in `GO_ROOM "WEST`. The procedure looks like this:

```

TO GO_ROOM :DIRECTION
  MAKE "DIRECTION GPROP :ROOM :DIRECTION
  IF :DIRECTION = "NOTHING THEN PRINT [YOU CAN'T GO THAT WAY.] STOP
  MAKE "ROOM :DIRECTION
  (PRINT [YOU ARE NOW IN THE] :ROOM)
END

```

You can just use these procedures as they are written, but for those who are curious, let's see how the `GO_ROOM` procedure works.

The first line uses `GPROP` (short for `Get Property`) to look up the room that is in the direction of the input word in the property list for the current room, stored in the variable `:ROOM`.

```
MAKE "DIRECTION GPROP :ROOM :DIRECTION
```

If you type `GPROP "ENTRY "NORTH`, `GPROP` looks up what room is north of `ENTRY`. The result is `"HALL`. So, the instruction above sets the value of `:DIRECTION` to `"HALL`.

If there isn't a property for the direction you type, you are told you can't go that way.

```
IF :DIRECTION = "NOTHING THEN PRINT [YOU CAN'T GO THAT WAY.] STOP
```

If there is a room in the direction of the input, then you set that new room to be the current room and tell the player where they are now.

```
MAKE "ROOM :DIRECTION
(PRINT [YOU ARE NOW IN THE] :ROOM)
```

Let's see if it works. After creating the `GO_ROOM` and the direction word procedures (`NORTH`, `SOUTH`, `EAST`, and `WEST`) in the Editor, click **Run** to execute the `SETUP` procedure.

If you are in the `ENTRY` and type `SOUTH`, what will you see? `YOU CAN'T GO THAT WAY.`

If you are in the `ENTRY` and type `NORTH`, what will you see? `YOU ARE NOW IN THE HALL.`

Don't worry if you don't yet fully understand the `GO_ROOM` procedure, as it is a bit complicated. You can just type it in and get it working.

After you have property lists for all the rooms and you can navigate to and from all the rooms (do lots of testing!), we'll add objects to find and explore.

Remember to save your program!

Come back in a couple of days to add more to your adventure!